```
> restart;
```

```
> with(LinearAlgebra);
```

[*&x*, *Add*, *Adjoint*, *BackwardSubstitute*, *BandMatrix*, *Basis*, *BezoutMatrix*, *BidiagonalForm*, *BilinearForm*, *CARE*, *CharacteristicMatrix*, *CharacteristicPolynomial*, *Column*, *ColumnDimension*, *ColumnOperation*, *ColumnSpace*, *CompanionMatrix*, *CompressedSparseForm*, *ConditionNumber*, *ConstantMatrix*, *ConstantVector*, *Copy*, *CreatePermutation*, *CrossProduct*, *DARE*, *DeleteColumn*, *DeleteRow*, *Determinant*, *Diagonal*, *DiagonalMatrix*, *Dimension*, *Dimensions*, *DotProduct*, *EigenConditionNumbers*, *Eigenvalues*, *Eigenvectors*, *Equal*, *ForwardSubstitute*, *FrobeniusForm*, *FromCompressedSparseForm*, *FromSplitForm*, *GaussianElimination*, *GenerateEquations*, *GenerateMatrix*, *Generic*, *GetResultDataType*, *GetResultShape*, *GivensRotationMatrix*, *GramSchmidt*, *HankelMatrix*, *HermiteForm*, *HermitianTranspose*, *HessenbergForm*, *HilbertMatrix*, *HouseholderMatrix*, *IdentityMatrix*, *IntersectionBasis*, *IsDefinite*, *IsOrthogonal*, *IsSimilar*, *IsUnitary*, *JordanBlockMatrix*, *JordanForm*, *KroneckerProduct*, *LA_Main*, *LUDecomposition*, *LeastSquares*, *LinearSolve*, *LyapunovSolve*, *Map*, *Map2*, *MatrixAdd*, *MatrixExponential*, *MatrixFunction*, *MatrixInverse*, *MatrixMatrixMultiply*, *MatrixNorm*, *MatrixPower*, *MatrixScalarMultiply*, *MatrixVectorMultiply*, *MinimalPolynomial*, *Minor*, *Modular*, *Multiply*, *NoUserValue*, *Norm*, *Normalize*, *NullSpace*, *OuterProductMatrix*, *Permanent*, *Pivot*, *PopovForm*, *ProjectionMatrix*, *QRDecomposition*, *RandomMatrix*, *RandomVector*, *Rank*, *RationalCanonicalForm*, *ReducedRowEchelonForm*, *Row*, *RowDimension*, *RowOperation*, *RowSpace*, *ScalarMatrix*, *ScalarMultiply*, *ScalarVector*, *SchurForm*, *SingularValues*, *SmithForm*, *SplitForm*, *StronglyConnectedBlocks*, *SubMatrix*, *SubVector*, *SumBasis*, *SylvesterMatrix*, *SylvesterSolve*, *ToeplitzMatrix*, *Trace*, *Transpose*, *TridiagonalForm*, *UnitVector*, *VandermondeMatrix*, *VectorAdd*, *VectorAngle*, *VectorMatrixMultiply*, *VectorNorm*, *VectorScalarMultiply*, *ZeroMatrix*, *ZeroVector*, *Zip*] **(1)**

Aproximare Pade in $x = 0$
apel R:=myPade(f,m,k,x)
f - functia
m,k - gradele
x - variabila

```
> myPade:=proc(f::procedure,m::nonnegint,k::nonnegint,x::name)
    description "Pade approximation";
    uses LinearAlgebra;
    local a,b,i,j,p,C,bv,bn,c,l,u;
    #Construct C matrix
    for p from 0 to m+k do
```

```
      c[p]:=coeftayl(f(x),x=0,p);
    end do;
    C:=ToeplitzMatrix([seq(c[m+i],i=-(k-1)..k-1)]);
    #construct rhs
    bv:=<seq(-c[m+i],i=1..k)>;
    bn:=LinearSolve(C,bv);
    b[0]:=1;
    for i from 1 to k do b[i]:=bn[i]; end do;
    #compute a
    if m > k then
      for i from k+1 to m do b[i]:=0; end do;
    end if;
    for j from 0 to m do
       a[j]:=sum(c[j-l]*b[l],l=0..j);
    end do;
    return sum(a[pp]*x^pp,pp=0..m)/sum(b[pp]*x^pp,pp=0..k);
  end proc;
```

myPade := **proc**(*f*::*procedure*, *m*::*nonnegint*, *k*::*nonnegint*, *x*::*name*) **(2)**

local *a*, *b*, *i*, *j*, *p*, *C*, *bv*, *bn*, *c*, *l*, *u*;

description "Pade approximation";

for *p* **from** 0 **to** $m+k$ **do** $c[p] := coeftayl(f(x), x=0, p)$ **end do**;

$C := LinearAlgebra:\text{-}ToeplitzMatrix([seq(c[m+i], i=-k+1..k-1)])$;

$bv := <seq(-c[m+i], i=1..k)>$;

$bn := LinearAlgebra:\text{-}LinearSolve(C, bv)$;

$b[0] := 1$;

for *i* **to** *k* **do** $b[i] := bn[i]$ **end do**;

if $k < m$ **then for** *i* **from** $k+1$ **to** *m* **do** $b[i] := 0$ **end do end if**;

for *j* **from** 0 **to** *m* **do** $a[j] := sum(c[j-l]*b[l], l=0..j)$ **end do**;

return $sum(a[pp]*x^pp, pp=0..m) / sum(b[pp]*x^pp, pp=0..k)$

end proc

```
> R22:=myPade(exp,2,2,x);
```

$$R22 := \frac{1 + \frac{1}{2}x + \frac{1}{12}x^2}{1 - \frac{1}{2}x + \frac{1}{12}x^2} \quad \textbf{(3)}$$

```
> with(numapprox);
```

[*chebdeg*, *chebmult*, *chebpade*, *chebsort*, *chebyshev*, *confracform*, *hermite_pade*, *hornerform*, **(4)**

infnorm, *laurent*, *minimax*, *pade*, *remez*]

```
> pade(exp(x),x=0,[2,2]);
```

$$\frac{1+\frac{1}{2}x+\frac{1}{12}x^2}{1-\frac{1}{2}x+\frac{1}{12}x^2} \tag{5}$$

```
> R33:=myPade(exp,3,3,x);
```

$$R33 := \frac{1+\frac{1}{2}x+\frac{1}{10}x^2+\frac{1}{120}x^3}{1-\frac{1}{2}x+\frac{1}{10}x^2-\frac{1}{120}x^3} \tag{6}$$

```
> pade(exp(x),x=0,[3,3]);
```

$$\frac{1+\frac{1}{2}x+\frac{1}{10}x^2+\frac{1}{120}x^3}{1-\frac{1}{2}x+\frac{1}{10}x^2-\frac{1}{120}x^3} \tag{7}$$

```
> R:=myPade(sin,2,2,t);
```

$$R := \frac{t}{1+\frac{1}{6}t^2} \tag{8}$$

```
> R:=myPade(sin,3,3,t);
```

$$R := \frac{t-\frac{7}{60}t^3}{1+\frac{1}{20}t^2} \tag{9}$$

```
> pade(sin(x),x=0,[3,3]);
```

$$\frac{-\frac{7}{60}x^3+x}{\frac{1}{20}x^2+1} \tag{10}$$

```
> pade(sin(x),x=0,[2,2]);
```

$$\frac{x}{\frac{1}{6}x^2+1} \tag{11}$$

```
> M:=(n,x)->convert(taylor(exp(x),x=0,n),polynom);
```

$$M := (n,x) \rightarrow convert\left(taylor\left(e^x, x=0, n\right), polynom\right) \tag{12}$$

```
> plot([exp(x),M(4,x),M(6,x),R22,R33],x=-3..3,legend=["exp","T4",
  "T6","R22","R33"]);
```




```
> plot([exp(x),M(4,x),M(6,x),R22,R33],x=-1..1,legend=["exp","T4",
  "T6","R22","R33"]);
```




```
> plot([abs(exp(x)-M(4,x)),abs(exp(x)-M(6,x)),abs(x-R22),abs(x-R33)
  ],x=-1..1,legend=["T4","T6","R22","R33"]);
```




```
> R122:=myPade(x->ln(1+x),2,2,x);
```

(13)

$$Rl22 := \frac{x + \frac{1}{2}x^2}{1 + x + \frac{1}{6}x^2} \tag{13}$$

```
> Rl32:=myPade(x->ln(1+x),3,2,x);
```

$$Rl32 := \frac{x + \frac{7}{10}x^2 + \frac{1}{30}x^3}{1 + \frac{6}{5}x + \frac{3}{10}x^2} \tag{14}$$

```
> pade(ln(1+t),t=0,[3,2]);
```

$$\frac{\frac{1}{30}t^3 + \frac{7}{10}t^2 + t}{1 + \frac{6}{5}t + \frac{3}{10}t^2} \tag{15}$$

```
> Rl31:=myPade(x->ln(1+x),3,1,x);
```

$$Rl31 := \frac{x + \frac{1}{4}x^2 - \frac{1}{24}x^3}{1 + \frac{3}{4}x} \tag{16}$$

```
> pade(ln(1+t),t=0,[3,1]);
```

$$\frac{-\frac{1}{24}t^3 + \frac{1}{4}t^2 + t}{1 + \frac{3}{4}t} \tag{17}$$

```
> f:=x->BesselJ(0,2*x);
```

$$f := x \rightarrow \mathrm{BesselJ}(0, 2x) \tag{18}$$

```
> taylor(f(x),x=0,10);
```

$$1 - x^2 + \frac{1}{4}x^4 - \frac{1}{36}x^6 + \frac{1}{576}x^8 + \mathrm{O}(x^{10}) \tag{19}$$

```
> ptB:=convert(%,polynom);
```

$$ptB := 1 - x^2 + \frac{1}{4}x^4 - \frac{1}{36}x^6 + \frac{1}{576}x^8 \tag{20}$$

```
> Rb22:=myPade(f,2,2,x);
```

$$Rb22 := \frac{1 - \frac{3}{4}x^2}{1 + \frac{1}{4}x^2} \tag{21}$$

```
> pade(f(x),x=0,[2,2]);
```

(22)

$$\frac{1-\frac{3}{4}x^2}{1+\frac{1}{4}x^2} \tag{22}$$

```
> Rb43:=myPade(f,4,4,x);
```

$$Rb43 := \frac{1-\frac{17}{20}x^2+\frac{79}{720}x^4}{1+\frac{3}{20}x^2+\frac{7}{720}x^4} \tag{23}$$

```
> pade(f(x),x=0,[4,4]);
```

$$\frac{1-\frac{17}{20}x^2+\frac{79}{720}x^4}{1+\frac{3}{20}x^2+\frac{7}{720}x^4} \tag{24}$$

```
> plot([BesselJ(0,2*x),ptB,Rb22,Rb43],x=-Pi..Pi,y=-5..5,legend=
  [typeset(J[0](x)),typeset(T[8](f)(x)), typeset('R[2,2]'), typeset
  ('R[4,3]')]);
```